

目 录

第一章 简易统计方法	(1)
1.1 直方图	(1)
1.1.1 基础知识	(1)
1.1.2 程序框图	(2)
1.1.3 程序清单	(2)
1.1.4 运行实例	(8)
1.2 排列图	(9)
1.2.1 基础知识	(9)
1.2.2 程序框图	(9)
1.2.3 程序清单	(9)
1.2.4 运行实例	(14)
1.3 散布图	(15)
1.3.1 基础知识	(15)
1.3.2 程序框图	(15)
1.3.3 程序清单	(15)
1.3.4 运行实例	(22)
第二章 计量值控制图	(23)
2.1 概述	(23)
2.1.1 计量值控制图的用途及格式	(23)
2.1.2 计量值控制图的原理及种类	(23)
2.1.3 计量值控制图的判别准则	(24)
2.1.4 计量值控制图的程序框图	(25)
2.2 平均值-极差控制图($\bar{x}-R$ 图)	(25)
2.2.1 基础知识	(25)
2.2.2 程序框图	(30)
2.2.3 程序清单	(30)
2.2.4 运行实例	(44)
2.3 中值-极差控制图($\bar{x}-R$ 图)	(48)
2.3.1 基础知识	(48)
2.3.2 程序框图	(49)
2.3.3 程序清单	(49)
2.3.4 运行实例	(64)

2.4 单值-移动极差控制图($x-R_s$ 图)	(67)
2.4.1 基础知识	(67)
2.4.2 程序框图	(68)
2.4.3 程序清单	(68)
2.4.4 运行实例	(81)
2.5 平均值-标准差控制图($\bar{x}-S$ 图)	(84)
2.5.1 基础知识	(84)
2.5.2 程序框图	(86)
2.5.3 程序清单	(86)
2.5.4 运行实例	(100)
第三章 计数值控制图	(104)
3.1 概述	(104)
3.1.1 计数值控制图的用途及格式	(104)
3.1.2 计数值控制图的原理及种类	(104)
3.1.3 计数值控制图的判别准则	(104)
3.1.4 计数值控制图的程序框图	(105)
3.2 不合格品率控制图(p 图)	(105)
3.2.1 基础知识	(105)
3.2.2 程序框图	(106)
3.2.3 程序清单	(106)
3.2.4 运行实例	(118)
3.3 不合格品数控制图(pn 图)	(121)
3.3.1 基础知识	(121)
3.3.2 程序框图	(121)
3.3.3 程序清单	(121)
3.3.4 运行实例	(132)
3.4 单位缺陷数控制图(u 图)	(135)
3.4.1 基础知识	(135)
3.4.2 程序框图	(136)
3.4.3 程序清单	(136)
3.4.4 运行实例	(148)
3.5 缺陷数控制图(c 图)	(150)
3.5.1 基础知识	(150)
3.5.2 程序框图	(151)
3.5.3 程序清单	(151)
3.5.4 运行实例	(162)

第四章 抽样检查	(166)
4.1 概述	(166)
4.1.1 抽样检查	(166)
4.1.2 抽样方案及其量度	(166)
4.2 计数标准型抽样方案	(167)
4.2.1 计数标准型一次抽样方案	(167)
4.2.2 计数标准型二次抽样方案	(172)
4.3 计数调整型抽样方案组	(176)
4.3.1 计数调整型一次抽样方案组	(177)
4.3.2 计数调整型二次抽样方案组	(183)
参考文献	(190)

第一章 简易统计方法

简易统计方法包括直方图、排列图、散布图、波动图、因果图和工序能力指数等，是生产现场经常使用、易于掌握的统计方法，在统计质量控制中其使用率占 80% 以上。

1.1 直方图

1.1.1 基础知识

直方图适用于对大量计量值数据进行整理加工，找出其统计规律，以便对总体的分布特征进行推断。直方图的作图程序包括收集数据、找出数据中的最大值 X_{max} 和最小值 X_{min} 、计算极差 R 、确定分组数 k 、确定组距 L 、计算各组上、下界限值和组中值、作频数分布表、作频数直方图、计算均值 \bar{x} 和标准差 S 、确定工序能力指数等几个步骤。

一般要求收集数据的个数 $n > 50$ ，理论上越多越好。分组数 k 既不宜太多，也不宜过少，组数太多会使各组的高度参差不齐，看不出明显的规律，组数过少会掩盖各组内数据的变化情况。确定上、下界限值时，应使最小值包含在第一组中，而最大值应包含在最后一组中，且应使数据观测值 x_i ($i = 1, 2, \dots, n$) 不落在上、下界限值上。

分组数 k 可以参照表 1.1 或图 1.1 确定，也可按如下公式近似计算，并取整。

$$k = 1 + 3.31 \lg n \quad (1.1)$$

$$\text{或 } k = 0.5 + \sqrt{n} \quad (1.2)$$

表 1.1

总数 n	分组数 k
50-100	6-10
100-250	7-12
250 以上	10-20




图 1.1

均值 \bar{x} 和标准差 S 是直方图的两个最基本的统计特征量，可按如下公式求得

$$\bar{x} = \frac{1}{n} \sum_{i=1}^n x_i \quad (1.3)$$

$$S = \sqrt{\frac{1}{n-1} \left[\sum_{i=1}^n x_i^2 - n \cdot \bar{x}^2 \right]} \quad (1.4)$$

当工序无偏时，工序能力指数 C_p ，

$$C_p = \frac{T}{6\sigma} \approx \frac{T}{6S} \quad (1.5)$$

上式中 T 代表公差大小。

当工序有偏时，修正的工序能力指数 C_{pk}

$$C_{pk} = \frac{T - 2\varepsilon}{6\sigma} \approx \frac{T - 2\varepsilon}{6S} \quad (1.6)$$

上式中 ε 为分布中心与公差带中心之间的偏移量的绝对值。

1.1.2 程序框图

直方图法的程序框图如图 1.2 所示。

1.1.3 程序清单

直方图法的程序清单如下：

```

10 REM 直方图主程序 编号 No: SQC.1F
20 PRINT "您想使用直方图作统计分析吗? y—是 n—否"
30 INPUT Y$
40 IF (Y$ <> "y") THEN PRINT "您的命令有误!":END
50 DIM X(800), FR(50), DI(50)
60 DEFINT I,J,K,N
70 PRINT "您的原始数据取自磁盘上的数据文件吗? y—是 n—否"
80 INPUT Y$
90 IF (Y$ <> "y") THEN 210
100 PRINT "您的数据文件名叫什么? 请键入文件名"
110 INPUT YY$
120 OPEN YY$ FOR INPUT AS #1
130 INPUT #1,UNIT$,WORKPICE$,MAN$,T$
135 INPUT #1,N
140 INPUT #1,TU,TL
150 INPUT #1,DX
160 FOR I=1 TO N
170 INPUT #1,X(I)
180 NEXT I
190 CLOSE #1
200 GOTO 485

```




图 1.2 直方图程序框图

```

210 PRINT "您的原始数据是键入的呢？还是读程序中的data语句获得的？1—键
      2—data"
220 INPUT B%
230 IF (B%=1) THEN 310
240 READ UNIT$,WORKPICE$,MAN$,T$
245 READ N
250 READ TU,TL
260 READ DX
270 FOR I=1 TO N
280 READ X(I)
290 NEXT I
300 GOTO 485
310 PRINT "请键入您的单位名：" :INPUT UNIT$
320 PRINT "请键入被测工件名：" :INPUT WORKPICE$
330 PRINT "请键入测量者姓名：" :INPUT MAN$
340 PRINT "请键入测量日期 (格式如 85-04-30.09)：" :INPUT T$
350 PRINT "请键入被测工件尺寸的公差上，下限：" :INPUT TU,TL
355 PRINT "请键入您的测量单位 dx=" :INPUT DX
360 PRINT "请键入测量数据的个数：" :INPUT N
370 PRINT "下面请键入原始数据 X(I)："
380 FOR I=1 TO N
390 PRINT "X(";I; :INPUT ")=";X(I)
410 PRINT "您刚才键入的数据需要修改吗？ y—需要 n—不需要 "
420 INPUT Y$
430 IF (Y$ <> "y") THEN 480
450 PRINT "X(";I;:INPUT ")=";X(I)
460 PRINT "还要修改吗？ y—要 n—不要 " :INPUT Y$
470 IF (Y$ ="y") THEN 450
480 NEXT I
485 LPRINT :LPRINT
490 LPRINT "单位名：" ;UNIT$
500 LPRINT "被测工件名：" ;WORKPICE$
510 LPRINT "被测工件的公差上，下限：" ;TU;"—" ;TL
520 LPRINT "测量者姓名：" ;MAN$
530 LPRINT "测量日期：" ;T$
540 LPRINT :LPRINT :LPRINT
550 LPRINT "原始数据表"

```

```

560 PICTURE$ = "-"
570 FOR I=1 TO 59
580 PICTURE$ = PICTURE$ + "-"
590 NEXT I
605 ZENG$ = " * "
610 FOR J=1 TO 200
625 ZENG$ = ZENG$ + " * "
630 NEXT J
635 LPRINT PICTURE$
640 FOR I=0 TO INT(N / 5+.05)-1
650 FOR J=1 TO 5
660 LPRINT USING "#####.###";X(5 * I+J);
665 LPRINT SPC(2);
670 NEXT J
680 LPRINT : NEXT I
690 LPRINT PICTURE$
700 XBAR=0:V=0
710 FOR I=1 TO N
720 XBAR=XBAR+X(I):V=V+X(I)2
730 NEXT I
740 XBAR=XBAR / N:V=SQR(ABS(V-N * XBAR2) / (N-1))
750 FOR I=1 TO N: FOR J=I TO N
760 IF (X(I) < X(J)) THEN 780
770 A1=X(J):X(J)=X(I):X(I)=A1
780 NEXT J
790 NEXT I
800 R=X(N)-X(1)
810 K=INT(SQR(N)+.5)
815 L=R / K
820 A1=X(1)-DX / 2
830 FOR I=0 TO K+1
840 DI(I)=A1+I * L
850 NEXT I
860 FOR I=1 TO K+1
870 FR(I)=0
880 FOR J=1 TO N
890 IF (X(J) < DI(I) AND X(J) > DI(I-1)) THEN FR(I)=FR(I)+1
900 NEXT J,I

```



```

910 LPRINT :LPRINT :LPRINT
920 LPRINT "组号:" 组 限 " 组中值 " 频数 " 直方图"
925 LPRINT PICTURE $
930 FOR I=1 TO K+1
940 LPRINT USING "####",I;
950 LPRINT USING "####.####";DI(I-1);
960 LPRINT "—";
970 LPRINT USING "####.####";DI(I);(DI(I)+DI(I-1))/2;
980 LPRINT USING "####";FR(I);
985 LPRINT TAB(38);"T";
990 KUN $ = MID $(ZENG $,1,FR(I))
1000 LPRINT KUN $
1010 NEXT I
1020 LPRINT PICTURE $
1030 LPRINT "合计";TAB(32);N
1040 LPRINT PICTURE $
1100 TM=(TU+TL)/2:T=TU-TL
1110 EP=ABS(XBAR-TM)
1120 CPK=(T-2*EP)/6/V
1130 CP=T/6/V
1140 IF (CPK<0) THEN CPK=0
1670 LPRINT "所测尺寸的公差中心 TM=";TM
1680 LPRINT "该尺寸的平均值 XBAR=";XBAR
1690 LPRINT "该尺寸的标准差 ";V
1700 LPRINT "该尺寸的分布中心与公差中心之偏移量 E=";EP
1710 LPRINT "该工序的工程能力指数 Cp=";CP
1720 LPRINT SPC(20);"Cpk=";CPK
1780 PRINT "您的原始数据需要存档吗? y—需要 n—不需要"
1790 INPUT Y $
1800 IF (Y $="y") THEN GOSUB 2400
1820 PRINT "您要继续使用直方图作统计分析吗? y—要 n—不要"
1830 INPUT Y $
1840 IF (Y $="y") THEN 70
1850 END
2400 REM 原始数据存档子程序
2410 TT $ =T $ : T $ ="b."+T $
2420 OPEN T $ FOR OUTPUT AS #1
2430 PRINT #1,UNIT $ CHR $(13) WORKPICE $ CHR $(13) MAN $ CHR

```

§ (13) TT §

```
2440 PRINT #1,N;TU;TL;DX
2450 FOR I=1 TO N
2460 PRINT #1,X(I)
2470 NEXT I
2480 CLOSE #1
2490 RETURN
2600 REM 主程序中固有数据
2610 DATA "华东工学院 83-研", "引信夹板厚度"
2620 DATA "曹坤增", "85-04-08.09"
2630 DATA 120 :
2640 DATA 4.220,4.200,0.001
2650 DATA 4.216,4.193,4.210,4.218,4.211,4.211,4.206,4.204
2660 DATA 4.208,4.206,4.223,4.207,4.211,4.206,4.213,4.203
2670 DATA 4.208,4.196,4.219,4.212,4.221,4.212,4.213,4.213
2680 DATA 4.203,4.216,4.204,4.202,4.207,4.206,4.208,4.216
2690 DATA 4.206,4.200,4.209,4.214,4.203,4.213,4.208,4.190
2700 DATA 4.214,4.208,4.209,4.206,4.202,4.208,4.216,4.207
2710 DATA 4.209,4.217,4.218,4.206,4.202,4.201,4.211,4.213
2720 DATA 4.211,4.216,4.211,4.208,4.213,4.208,4.204,4.211
2730 DATA 4.202,4.200,4.222,4.209,4.208,4.216,4.202,4.219
2740 DATA 4.197,4.208,4.211,4.219,4.226,4.199,4.206,4.204
2750 DATA 4.203,4.209,4.224,4.214,4.218,4.214,4.207,4.207
2760 DATA 4.194,4.208,4.208,4.211,4.218,4.206,4.210,4.199
2770 DATA 4.211,4.206,4.216,4.221,4.211,4.205,4.198,4.213
2780 DATA 4.208,4.208,4.211,4.214,4.206,4.213,4.214,4.201
2790 DATA 4.203,4.207,4.209,4.206,4.201,4.207,4.214,4.209
```

1.1.4 运行实例

以直方图法程序中 DATA 语句所列的数据运行, 结果如下:

单位名: 华东工学院 83-研

被测工件名: 引信夹板厚度

被测工件的公差上, 下限: 4.22 — 4.2

测量者姓名: 曹坤增

测量日期: 85-04-08.09

原始数据表

4.216	4.193	4.210	4.218	4.211
4.211	4.206	4.204	4.208	4.206
4.223	4.207	4.211	4.206	4.213
4.203	4.208	4.196	4.219	4.212
4.221	4.212	4.213	4.213	4.203
4.216	4.204	4.202	4.207	4.206
4.208	4.216	4.206	4.200	4.209
4.214	4.203	4.213	4.208	4.190
4.214	4.208	4.209	4.206	4.202
4.208	4.216	4.207	4.209	4.217
4.218	4.206	4.202	4.201	4.211
4.213	4.211	4.216	4.211	4.208
4.213	4.208	4.204	4.211	4.202
4.200	4.222	4.209	4.208	4.216
4.202	4.219	4.197	4.208	4.211
4.219	4.226	4.199	4.206	4.204
4.203	4.209	4.224	4.214	4.218
4.214	4.207	4.207	4.194	4.208
4.208	4.211	4.218	4.206	4.210
4.199	4.211	4.206	4.216	4.221
4.211	4.205	4.198	4.213	4.208
4.208	4.211	4.214	4.206	4.213
4.214	4.201	4.203	4.207	4.209
4.206	4.201	4.207	4.214	4.209

组号	组 限	组中值	频数	直方图
1	4.189--	4.193	4.191	1 I*
2	4.193--	4.196	4.194	3 I***
3	4.196--	4.199	4.198	4 I****
4	4.199--	4.203	4.201	10 I*****
5	4.203--	4.206	4.204	10 I*****
6	4.206--	4.209	4.207	40 I*****
7	4.209--	4.212	4.211	16 I*****
8	4.212--	4.216	4.214	15 I*****
9	4.216--	4.219	4.217	12 I*****
10	4.219--	4.222	4.221	6 I*****
11	4.222--	4.225	4.224	2 I**
12	4.225--	4.229	4.227	1 I*
合计			120	

所测尺寸的公差中心 $TM = 4.21$

该尺寸的平均值 $XBAR = 4.209033$

该尺寸的标准差: $6.869268E-03$

该尺寸的分布中心与公差中心之偏移量 $E = 9.670258E-04$

该工序的工程能力指数 $Cp = .4852526$

$Cpk = .4383274$

1.2 排列图

1.2.1 基础知识

排列图可以帮助我们找出影响产品质量的主要因素, 确定需要改进的关键项目。实践证明, 集中精力将主要因素的影响减半比消灭次要因素的影响要收效显著, 而且容易得多。所以应当选取排列图前 1-2 项因素作为质量改进的目标。排列图的作图程序包括确定分析的对象、问题分类的项目、收集与整理数据、计算频数 f_i 、频率 p_i 和累计频率 F_i 、作图、确定主要影响因素、影响因素和次要因素等几个步骤。

频数 f_i 是指每个项目发生的数量, 频率 p_i 和累计频率 F_i 如下计算

$$p_i = f_i / n \quad (1.7)$$

$$F_i = \sum_{j=1}^i p_j \quad (1.8)$$

上式中 n 代表所有项目发生的总数。

1.2.2 程序框图

排列图法的程序框图如图 1.3 所示。

1.2.3 程序清单

排列图法的程序清单如下:

10 REM 排列图主程序 编号 No:SQC.10F




图 1.3 排列图程序框图

```

20 PRINT "您想使用排列图? y—是 n—否"
30 INPUT Y$
40 IF (Y$ <> "y") THEN PRINT "您的命令有误!":END
50 DIM NF(100),PN(100)
60 DEFINT I,J,K,N
70 PRINT "您的原始数据取自磁盘上的数据文件吗? y—是 n—否"
80 INPUT Y$
90 IF (Y$ <> "y") THEN 210
100 PRINT "您的数据文件名叫什么? 请键入文件名"
110 INPUT YY$
120 OPEN YY$ FOR INPUT AS #1
130 INPUT #1,UNIT$,WORKPICE$,MAN$,T$
135 INPUT #1,K,N
160 FOR I=1 TO N
170 INPUT #1,CA$(I),NF(I)
180 NEXT I
190 CLOSE #1
200 GOTO 485
210 PRINT "您的原始数据是键入的呢? 还是读程序中的data语句获得的? 1—键
2--data"
220 INPUT B%
230 IF (B%=1) THEN 310
240 READ UNIT$,WORKPICE$,MAN$,T$
245 READ K,N
270 FOR I=1 TO N
280 READ CA$(I),NF(I)
290 NEXT I
300 GOTO 485
310 PRINT "请键入您的单位名: " :INPUT UNIT$
320 PRINT "请键入被测工件名: " :INPUT WORKPICE$
330 PRINT "请键入测量者姓名: " :INPUT MAN$
340 PRINT "请键入测量日期 (格式如 85-04-30.09):" :INPUT T$
360 PRINT "请键入样本大小(N):" :INPUT N
370 PRINT "请键入缺陷项目数(K):" :INPUT N
375 PRINT "下面请键入原始数据:"
380 FOR I=1 TO N
390 PRINT "缺陷名称 CA$(I): INPUT ")=";CA$(I)

```

```

395 PRINT "缺陷个数 NF("J: INPUT ")=";NF(I)
410 PRINT "您刚才键入的这组数据需要修改吗? y--需要 n--不需要"
420 INPUT Y$
430 IF (Y$ <> "y") THEN 480
450 PRINT "缺陷名称 CA$("J: INPUT ")=";CA$(I)
455 PRINT "缺陷个数 NF("J: INPUT ")=";NF(I)
460 PRINT "还要修改吗? y--要 n--不要":INPUT Y$
470 IF (Y$ = "y") THEN 450
480 NEXT I
485 LPRINT :LPRINT
490 LPRINT "单位名: ";UNIT$
500 LPRINT "被测工件名: ";WORKPIECE$
520 LPRINT "测量者姓名: ";MAN$
530 LPRINT "测量日期: ";T$
540 LPRINT :LPRINT :LPRINT
550 LPRINT "原始数据表"
560 PICTURE$ = "-"
570 FOR I=1 TO (8 * N+18)
580 PICTURE$ = PICTURE$ + "-"
590 NEXT I
600 CH$ = " "
605 ZENG$ = " * "
610 FOR J=1 TO 65
620 CH$ = CH$ + " "
625 ZENG$ = ZENG$ + " * "
630 NEXT J
635 LPRINT PICTURE$
640 LPRINT "缺陷项目","频数"
650 LPRINT PICTURE$
655 TOTAL=0
660 FOR I=1 TO N
670 LPRINT CA$(I),NF(I)
675 TOTAL=TOTAL+NF(I)
680 NEXT I
690 LPRINT PICTURE$
700 LPRINT "合计",TOTAL
710 LPRINT PICTURE$
720 LPRINT :LPRINT :LPRINT

```

```

730 LPRINT "排列图", "样本大小";K,T$
735 LPRINT PICTURE$:LPRINT :LPRINT
740 FOR I=1 TO N
750 PN(I)=NF(I)/TOTAL
760 NEXT I
770 FOR I=1 TO N-1 : FOR J=1 TO N-1
780 IF (NF(I)>=NF(J)) THEN 810
790 A1=NF(J):NF(J)=NF(I):NF(I)=A1
795 A2=PN(J):PN(J)=PN(I):PN(I)=A2
800 AA$=CA$(J):CA$(J)=CA$(I):CA$(I)=AA$
810 NEXT J
815 NEXT I
825 TOTAL=0
830 FOR I=1 TO N
835 TOTAL=TOTAL+PN(I)
840 LPRINT CA$(I),
850 KUN$=MID$(ZENG$,I,NF(I))
860 LPRINT KUN$;
870 LPRINT NF(I);PN(I);(":"TOTAL;")"
890 NEXT I
900 LPRINT :LPRINT
910 LPRINT PICTURE$
920 LPRINT WORKPICE$,UNIT$;MAN$
930 PRINT "您的原始数据需要存档吗? y—需要 n—不需要"
940 INPUT Y$
950 IF (Y$="y") THEN GOSUB 2400
960 PRINT "您要继续使用排列图吗? y—要 n—不要"
970 INPUT Y$
980 IF (Y$="y") THEN 70
990 END
2400 REM 原始数据存档子程序
2410 TT$=T$:T$="b:"+T$
2420 OPEN T$ FOR OUTPUT AS #1
2430 PRINT #1,UNIT$ CHR$(13) WORKPICE$ CHR$(13) MAN$ CHR
$(13) TT$
2440 PRINT #1,K;N
2450 FOR I=1 TO N
2460 PRINT #1,CA$(I) CHR$(13) NF(I)

```



```

2470 NEXT I
2480 CLOSE #1
2490 RETURN
2500 DATA "华东工学院 83-研", "引信零件"
2510 DATA "曹坤增", "85-6-11.16"
2520 DATA 10000, 6
2530 DATA "疵点", 41, "气孔", 18, "未充满", 13
2540 DATA "形状不佳", 10, "尺寸超差", 6
2550 DATA "其他", 7

```

1.2.4 运行实例

以排列图法程序中 DATA 语句所列的数据运行, 结果如下:

单位名称: 华东工学院 83-研

被测工件名: 引信零件

测量者姓名: 曹坤增

测量日期: 85-6-11.16

原始数据表

缺陷项目	频数
疵点	41
气孔	18
未充满	13
形状不佳	10
尺寸超差	6
其他	7
合计	95

排列图 样本大小: 10000 85-6-11.16

```

疵点 ***** 41 .431579 ( .431579 )
气孔 ***** 18 .1894737 ( .6210526 )
未充满 ***** 13 .1368421 ( .7578948 )
形状不佳 ***** 10 .1052632 ( .8631579 )
尺寸超差 ***** 6 .0631579 ( .9263159 )
其他 ***** 7 7.363421E-02 ( 1 )

```

引信零件 华东工学院83-研曹坤增